# ELEC6027 - VLSI Design Project Programmers Guide

Team R4

Henry Lovett (hl13g10)

Ashey J. Robinson (ajr2g10)

Martin Wearn (mw20g10)

Anusha R. Reddy (arr1g13)

Course Tutor: Mr B. Iain McNally

 $2^{nd}$  May, 2014

# Contents

1	$\operatorname{Intr}$	oduction 7
	1.1	Architecture
	1.2	Register Description
<b>2</b>	Inst	ruction Set 11
	2.1	General Instruction Formatting
	2.2	ADD
	2.3	ADDI
	2.4	ADDIB
	2.5	ADC
	2.6	ADCI
	2.7	NEG
	2.8	SUB
	2.9	SUBI
	2.10	SUBIB
	2.11	SUC
	2.12	SUCI
	2.13	CMP
	2.14	CMPI
	2.15	AND
	2.16	OR
		XOR
	2.18	NOT
	2.19	NAND
	2.20	NOR
	2.21	LSL
	2.22	LSR
	2.23	ASR
	2.24	LDW
	2.25	STW
		LUI
	_	LLI
	2.28	
		BNE
	2.20	

	2.31	BLT
	2.32	BGE
	2.33	BWL
	2.34	RET
	2.35	JMP
	2.36	PUSH
	2.37	POP
	2.38	RETI 50
		ENAI
	2.40	DISI
	2.41	STF
	2.42	LDF
3	Pro	gramming Tips 55
	3.1	Branching
	3.2	Looping
	3.3	Stack Pointer Usage
	3.4	Sub routine calling convention
	3.5	Interrupt Service Routines
4	Ass	embler 63
	4.1	Instruction Formatting
	4.2	Assembler Directives
	4.3	Running The Assembler
	4.4	Error Messages
5	Pro	$_{ m grams}$
	5.1	Multiply
	5.2	Factorial
	5.3	Random
	5.4	Interrupt
6	Sim	ulation 75
	6.1	Running the simulations
	6.2	Serial Data
	6.3	Run Time
	6.4	Simulation

AMURAI:
Prog
grammers (
Guide

$\mathbf{A}$	Cod	e Listings	<b>7</b> 9
	A.1	Multiply	79
	A.2	Factorial	80
	A.3	Random	81
	A.4	Interrupt	82
ъ	Ъ		0.
В	Dec	aration of Work	87

## 1 Introduction

This is the Programmers Guide for the processor designed by Team R4 in the VLSI Design Project, ELEC6027.

The processor is called Samurai - Sixteen bit ARM and MIPS Unified RISC Architecture with Interrupts. It is a sixteen bit general purpose Von Neumann processor. Samurai implements a custom Instruction Set.

This guide documents the architecture and instruction set. In addition, four example programs are given along with instructions of the use of the Assembler. Finally, the simulation environment is explained, giving examples of how to run a program.

#### 1.1 Architecture

Figure 1 shows the datapath architecture of the SAMURAI processor. The controller has been omitted along with all control signals. The exception is the status register is shown for data flow as this utilises the System Bus. Instruction decoding is also not shown for clarity. All registers, buses and multiplexors are 16 bits in length unless otherwise stated.

## 1.2 Register Description

The Samurai processor has twelve registers in total, all are 16 bits wide. These is a program counter, instruction register, link register, ALU output register and 8 general purpose registers. Each register is described below, along with any conventions.

General Purpose Registers The register block consists of eight General Purpose Registers (GPRs). There is no dummy register. By convention, Register 7 is used as the stack pointer. It is used by stack and interrupt instructions such as push, pop, store and load flags and return from interrupt.

Link Register The Link Register is used to store the return address of the caller function. However, it is not a part of the General Purpose register file. The link register is used by the branch with link and return from subroutine instructions. In these, the program counter is stored to or set by the link register. The link register can also be pushed or popped to/from the stack.




Figure 1: The Architecture diagram of the Samurai processor.

**Program Counter** The program counter is used to access the current instruction. It can be set by the result of an ALU operation, the link register, a value on the stack or a predefined constant used for interrupts. Branch instructions are the main modifier of the program counter. By default, all instructions increment the Program Counter by one to progress the operation of the program. This is not an addressable register and it's functionality is utilised by the control unit only.

**Instruction Register** The instruction register contains the currently executed instruction. This can only be set from the main memory by use of the Program Counter as the address to main memory. It is not addressable and it's function is utilised by the control unit.

**AluOut** The AluOut register is used to hold a value on the output of the ALU. It is used by memory access instructions and is not addressable.

## 2 Instruction Set

The complete instruction set architecture includes a number of instructions for performing calculations on data, memory access, transfer of control within a program and interrupt handling.

All instructions implemented by this architecture fall into one of 6 groups, categorised as follows:

- Data Manipulation Arithmetic, Logical, Shifting
- Byte Immediate Arithmetic, Byte Load
- Data Transfer Memory Access
- Control Transfer (Un)conditional Branching
- Stack Operations Push, Pop
- Interrupts Enabling, Status Storage, Returning

There is only one addressing mode associated with each instruction, generally following these groupings:

- Data Manipulation Register-Register, Register-Immediate
- Byte Immediate Register-Immediate
- Data Transfer Base Plus Offset
- Control Transfer PC Relative, Register-Indirect, Base Plus Offset
- Stack Operations Register-Indirect Preincrement/Postdecrement
- Interrupts Register-Indirect Preincrement/Postdecrement

## 2.1 General Instruction Formatting

	Instruction Type	Sub-Type	15	14	13	12	11	10	9 8	7 6 5	4 3 2	1 0
A1	Data Manipulation	Register		Or	oco	ما		]	Rd	Ra	Rb	X X
A2	Data Manipulation	Immediate		O <sub>I</sub>	,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,,	10		]	Rd	Ra	imm4	l/5
В	Byte Immediate		$O_{\rm I}$	oco	de		]	Rd		imm8		
С	Data Transfer	0	LS	0	0	0	]	Rd	Ra	imm	ı5	
D1	Control Transfer	Others	1	1	1	1	0	C	ond.		imm8	
D2	Jump		1	1	1	1	U	0	ond.	Ra	imm	ı5
Е	Stack Operations		0	U	0	0	1	L	X X	Ra	0 0 0	0 1

0 0 1 | ICond. | 1 1 1 | X X X X X

#### **Instruction Field Definitions**

F | Interrupts

Opcode: Operation code as defined for each instruction

Rd: Destination Register

Ra: Source register 1

Rb: Source register 2

immN: Immediate value of length N

Cond.: Branching condition code as defined for branch instructions

ICond.: Interrupt instruction code as defined for interrupt instructions

LS: 0=Load Data, 1=Store Data

U: 1=PUSH, 0=POP

L: 1=Use Link Register, 0=Use GPR

## Pseudocode Notation

Symbol	Meaning
<del></del>	Assignment
Result[x]	Bit $x$ of result
Ra[x: y]	Bit range from $x$ to $y$ of register Ra
<	Numerically less than
>	Numerically greater than
<<	Logical shift left
>>	Logical shift right
>>>	Arithmetic shift right
Mem[val]	Data at memory location with address $val$
$\{x, y\}$	Concatenation of $x$ and $y$ to form a 16-bit value
!	Bitwise Negation

Use of the word UNPREDICTABLE indicates that the resultant flag value after operation execution will not be indicative of the ALU result. Instead its value will correspond to the result of an undefined arithmetic operation and as such should not be used.

2.2 ADD Add Word

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0	0	0	1	0		Rd			Ra			Rb		X	X

### **Syntax**

ADD Rd, Ra, Rb

eg. ADD R5, R3, R2

## Operation

$$Rd \leftarrow Ra + Rb$$

$$N \leftarrow if (Result < 0) then 1, else 0$$

$$Z \leftarrow if (Result = 0) then 1, else 0$$

$$V \leftarrow if (Ra>0 \text{ and } Rb>0 \text{ and } Result<0) \text{ or}$$

(Ra<0 and Rb<0 and Result>0) then 1, else 0

$$C \leftarrow if (Result > 2^{16} - 1) or$$

(Result 
$$< -2^{16}$$
) then 1, else 0

## Description

The 16-bit word in GPR[Ra] is added to the 16-bit word in GPR[Rb] and the result is placed into GPR[Rd].

 ${\bf Addressing\ Mode:\ Register-Register}.$ 

#### 2.3 ADDI

## Add Immediate

#### **Format**

15	5	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0	)	0	1	1	0	-	Rd			Ra			iı	mm	5	

## **Syntax**

ADDI Rd, Ra, #imm5

eg. ADDI R5, R3, #7

## Operation

$$Rd \leftarrow Ra + imm5$$

$$N \leftarrow if (Result < 0) then 1, else 0$$

$$Z \leftarrow if (Result = 0) then 1, else 0$$

$$V \leftarrow if (Ra>0 \text{ and } \#imm5>0 \text{ and } Result<0) \text{ or }$$

(Ra<0 and #imm5<0 and Result>0) then 1, else 0

$$C \leftarrow if (Result > 2^{16} - 1) or$$

(Result 
$$< -2^{16}$$
) then 1, else 0

## Description

The 16-bit word in GPR[Ra] is added to the sign-extended 5-bit value given in the instruction and the result is placed into GPR[Rd].

#### **ADDIB** 2.4

## Add Immediate Byte

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0	0	0	1	1	-	Rd					im	m8			

### **Syntax**

ADDIB Rd, #imm8

eg. ADDIB R5, #93

## Operation

Rd 
$$\leftarrow$$
 Rd + imm8  
N  $\leftarrow$  if (Result < 0) then 1, else 0  
Z  $\leftarrow$  if (Result = 0) then 1, else 0  
V  $\leftarrow$  if (Rd>0 and #imm8>0 and Result<0) or  
(Rd<0 and #imm8<0 and Result>0) then 1, else 0  
C  $\leftarrow$  if (Result >  $2^{16} - 1$ ) or

(Result  $< -2^{16}$ ) then 1, else 0

## Description

The 16-bit word in GPR[Rd] is added to the sign-extended 8-bit value given in the instruction and the result is placed into GPR[Rd].

#### 2.5 ADC

## Add Word With Carry

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0	0	1	0	0		Rd			Ra			Rb		X	X

## **Syntax**

ADC Rd, Ra, Rb

eg. ADC R5, R3, R2

## Operation

$$Rd \leftarrow Ra + Rb + C$$
  
 $N \leftarrow if (Result < 0) then 1, else 0$ 

$$Z \leftarrow if (Result = 0) then 1, else 0$$

V 
$$\leftarrow$$
 if (Ra>0 and (Rb+CFlag)>0 and Result<0) or

(Ra<0 and (Rb+CFlag)<0 and Result>0) then 1, else 0 
$$\,$$

C 
$$\leftarrow$$
 if (Result  $> 2^{16} - 1$ ) or   
(Result  $< -2^{16}$ ) then 1, else 0

## Description

The 16-bit word in GPR[Ra] is added to the 16-bit word in GPR[Rb] with the added carry in set according to the Carry flag from previous operation. The result is then placed into GPR[Rd].

#### **ADCI** 2.6

## Add Immediate With Carry

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0	0	1	0	1		Rd			Ra			i	mm	5	

### **Syntax**

ADCI Rd, Ra, #imm5

eg. ADCI R5, R4, #7

## Operation

Rd 
$$\leftarrow$$
 Ra + imm5 + C  
N  $\leftarrow$  if (Result < 0) then 1, else 0  
Z  $\leftarrow$  if (Result = 0) then 1, else 0  
V  $\leftarrow$  if (Ra>0 and (#imm5+CFlag)>0 and Result<0) or  
(Ra<0 and (#imm5+CFlag)<0 and Result>0) then 1, else 0  
C  $\leftarrow$  if (Result >  $2^{16} - 1$ ) or  
(Result <  $-2^{16}$ ) then 1, else 0

## Description

The 16-bit word in GPR[Ra] is added to the sign-extended 5-bit value given in the instruction with carry in set according to the Carry flag from previous operation. The result is then placed into GPR[Rd].

## 2.7 NEG

# Negate Word

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1	0	1	0		Rd			Ra		X	X	X	X	X

## Syntax

NEG Rd, Ra

eg. NEG R5, R3

## Operation

$$\mathrm{Rd} \leftarrow 0$$
 -  $\mathrm{Ra}$ 

$$N \leftarrow if (Result < 0) then 1, else 0$$

$$Z \leftarrow if (Result = 0) then 1, else 0$$

$$V \leftarrow 0$$

$$\mathbf{C} \leftarrow \mathbf{0}$$

## Description

The 16-bit word in GPR[Ra] is subtracted from zero and the result is placed into GPR[Rd].

2.8 SUB
Format

## Subtract Word

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0	1	0	1	0		Rd			Ra			Rb		X	X

## **Syntax**

SUB Rd, Ra, Rb

eg. SUB R5, R3, R2

## Operation

$$\mathrm{Rd} \leftarrow \mathrm{Ra} - \mathrm{Rb}$$

$$N \leftarrow if (Result < 0) then 1, else 0$$

$$Z \leftarrow if (Result = 0) then 1, else 0$$

$$V \leftarrow if (Ra>0 \text{ and } Rb>0 \text{ and } Result<0) \text{ or}$$

(Ra<0 and Rb<0 and Result>0) then 1, else 0 
$$\,$$

$$C \leftarrow if (Result > 2^{16} - 1) or$$

(Result 
$$< -2^{16}$$
) then 1, else 0

## Description

The 16-bit word in GPR[Rb] is subtracted from the 16-bit word in GPR[Ra] and the result is placed into GPR[Rd].

#### 2.9 SUBI

## Subtract Immediate

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0	1	1	1	0		Rd			Ra			iı	mm	5	

## **Syntax**

SUBI Rd, Ra, #imm5

eg. SUBI R5, R3, #7

## Operation

$$Rd \leftarrow Ra - imm5$$

$$N \leftarrow if (Result < 0) then 1, else 0$$

$$Z \leftarrow if (Result = 0) then 1, else 0$$

$$V \leftarrow if \ (Ra{>}0 \ and \ \#imm5{>}0 \ and \ Result{<}0)$$
 or

(Ra<0 and #imm5<0 and Result>0) then 1, else 0

$$C \leftarrow if (Result > 2^{16} - 1) or$$

(Result 
$$< -2^{16}$$
) then 1, else 0

## Description

The sign extended 5-bit value given in the instruction is subtracted from the 16-bit word in GPR[Ra] and the result is placed into GPR[Rd].

#### 2.10 **SUBIB**

## Subtract Immediate Byte

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0	1	0	1	1	-	Rd					im	m8			

### **Syntax**

SUBIB Rd, #imm8

eg. SUBIB R5, #93

## Operation

$$Rd \leftarrow Rd - imm8$$

$$N \leftarrow if (Result < 0) then 1, else 0$$

$$Z \leftarrow if (Result = 0) then 1, else 0$$

$$V \leftarrow if \; (Rd{>}0 \; and \; \#imm8{>}0 \; and \; Result{<}0) \; or$$

(Rd<0 and 
$$\#imm8<0$$
 and Result>0) then 1, else 0

$$C \leftarrow \text{if (Result} > 2^{16} - 1) \text{ or}$$

(Result 
$$< -2^{16}$$
) then 1, else 0

## Description

The 8-bit immediate value given in the instruction is subtracted from the 16-bit word in GPR[Rd] and the result is placed into GPR[Rd].

#### 2.11 SUC

## Subtract Word With Carry

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0	1	1	0	0		Rd			Ra			Rb		X	X

## **Syntax**

SUC Rd, Ra, Rb

eg. SUC R5, R3, R2

## Operation

$$Rd \leftarrow Ra - Rb - C$$

$$N \leftarrow if (Result < 0) then 1, else 0$$

$$Z \leftarrow if (Result = 0) then 1, else 0$$

$$V \leftarrow if \; (Ra{>}0 \; and \; (Rb{-}CFlag){>}0 \; and \; Result{<}0) \; or \;$$

$$C \leftarrow if (Result > 2^{16} - 1) or$$

(Result 
$$< -2^{16}$$
) then 1, else 0

## Description

The 16-bit word in GPR[Rb] is subtracted from the 16-bit word in GPR[Rb] with the subtracted carry in set according to the Carry flag from previous operation. The result is then placed into GPR[Rd].

#### **SUCI** 2.12

## **Subtract Immediate With Carry**

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0	1	1	0	1		Rd			Ra			i	mm	5	

## **Syntax**

SUCI Rd, Ra, #imm5

eg. SUCI R5, R4, #7

## Operation

Rd 
$$\leftarrow$$
 Ra - imm5 - C  
N  $\leftarrow$  if (Result  $<$  0) then 1, else 0  
Z  $\leftarrow$  if (Result  $=$  0) then 1, else 0  
V  $\leftarrow$  if (Ra>0 and (#imm5-CFlag)>0 and Result<0) or  
(Ra<0 and (#imm5-CFlag)<0 and Result>0) then 1, else 0  
C  $\leftarrow$  if (Result  $>$  2<sup>16</sup>  $-$  1) or  
(Result  $<$   $-$  2<sup>16</sup>) then 1, else 0

## Description

The 5-bit immediate value in instruction is subtracted from the 16-bit word in GPR[Ra] with the subtracted carry in set according to the Carry flag from previous operation. The result is then placed into GPR[Rd].

## 2.13 CMP

## Compare Word

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0	0	1	1	1	X	X	X		Ra			Rb		X	X

## Syntax

CMP Ra, Rb

eg. CMP R3, R2

## Operation

 $N \leftarrow if (Result < 0) then 1, else 0$ 

 $Z \leftarrow if (Result = 0) then 1, else 0$ 

 $V \leftarrow if (Ra>0 \text{ and } Rb>0 \text{ and } Result<0) \text{ or }$ 

(Ra<0 and Rb<0 and Result>0) then 1, else 0

 $\mathbf{C} \leftarrow \text{if (Result} > 2^{16} - 1) \text{ or }$ 

(Result  $< -2^{16}$ ) then 1, else 0

## Description

The 16-bit word in GPR[Rb] is subtracted from the 16-bit word in GPR[Ra] and the status flags are updated without saving the result.

#### 2.14 **CMPI**

## Compare Immediate

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0	1	1	1	1	X	X	X		Ra			i	mm	5	

## **Syntax**

CMPI Ra, #imm5

eg. CMPI R3, #7

## Operation

$$N \leftarrow if (Result < 0) then 1, else 0$$

$$Z \leftarrow if (Result = 0) then 1, else 0$$

$$V \leftarrow if (Ra>0 \text{ and } \#imm5>0 \text{ and } Result<0) \text{ or }$$

(Ra<0 and #imm5<0 and Result>0) then 1, else 0

$$C \leftarrow if (Result > 2^{16} - 1) or$$

(Result 
$$< -2^{16}$$
) then 1, else 0

#### Description

The sign extended 5-bit value given in the instruction is subtracted from the 16-bit word in GPR[Ra] and the status flags are updated without saving the result.

#### 2.15 AND

# Logical AND

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	0	0	0	0		Rd			Ra			Rb		X	X

## **Syntax**

AND Rd, Ra, Rb

eg. AND R5, R3, R2

## Operation

 $Rd \leftarrow Ra \text{ AND } Rb$ 

 $N \leftarrow if (Result < 0) then 1, else 0$ 

 $Z \leftarrow if (Result = 0) then 1, else 0$ 

 $V \leftarrow UNPREDICTABLE$ 

 $C \leftarrow UNPREDICTABLE$ 

## Description

The logical AND of the 16-bit words in GPR[Ra] and GPR[Rb] is performed and the result is placed into GPR[Rd].

2.16 OR

Logical OR

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	0	0	0	1	-	Rd			Ra			Rb		X	X

## Syntax

OR Rd, Ra, Rb

eg. OR R5, R3, R2

## Operation

 $Rd \leftarrow Ra \; \mathtt{OR} \; Rb$ 

 $N \leftarrow if (Result < 0) then 1, else 0$ 

 $Z \leftarrow if (Result = 0) then 1, else 0$ 

 $V \leftarrow UNPREDICTABLE$ 

 $\mathbf{C} \leftarrow \mathbf{UNPREDICTABLE}$ 

#### Description

The logical OR of the 16-bit words in GPR[Ra] and GPR[Rb] is performed and the result is placed into GPR[Rd].

## 2.17 XOR

# Logical XOR

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	0	0	1	1		Rd			Ra			Rb		X	X

### **Syntax**

XOR Rd, Ra, Rb

eg. XOR R5, R3, R2

## Operation

 $Rd \leftarrow Ra XOR Rb$ 

 $N \leftarrow if (Result < 0) then 1, else 0$ 

 $Z \leftarrow if (Result = 0) then 1, else 0$ 

 $V \leftarrow UNPREDICTABLE$ 

 $C \leftarrow UNPREDICTABLE$ 

## Description

The logical XOR of the 16-bit words in GPR[Ra] and GPR[Rb] is performed and the result is placed into GPR[Rd].

#### 2.18 NOT

# Logical NOT

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	0	0	1	0		Rd			Ra		X	X	X	X	X

## Syntax

NOT Rd, Ra

eg. NOT R5, R3

## Operation

 $Rd \leftarrow \texttt{NOT}\ Ra$ 

 $N \leftarrow if (Result < 0) then 1, else 0$ 

 $Z \leftarrow if (Result = 0) then 1, else 0$ 

 $V \leftarrow UNPREDICTABLE$ 

 $C \leftarrow UNPREDICTABLE$ 

## Description

The logical NOT of the 16-bit word in GPR[Ra] is performed and the result is placed into  $\operatorname{GPR}[\operatorname{Rd}]$ .

#### 2.19 NAND

# Logical NAND

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	0	1	1	0	-	Rd			Ra			Rb		X	X

## **Syntax**

NAND Rd, Ra, Rb

eg. NAND R5, R3, R2

## Operation

 $Rd \leftarrow Ra \text{ NAND } Rb$ 

 $N \leftarrow if (Result < 0) then 1, else 0$ 

 $Z \leftarrow if (Result = 0) then 1, else 0$ 

 $V \leftarrow UNPREDICTABLE$ 

 $C \leftarrow UNPREDICTABLE$ 

## Description

The logical NAND of the 16-bit words in GPR[Ra] and GPR[Rb] is performed and the result is placed into GPR[Rd].

#### 2.20 NOR

## Logical NOR

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	0	1	1	1		Rd			Ra			Rb		X	X

## **Syntax**

NOR Rd, Ra, Rb

eg. NOR R5, R3, R2

## Operation

 $Rd \leftarrow Ra \ \texttt{NOR} \ Rb$ 

 $N \leftarrow if (Result < 0) then 1, else 0$ 

 $Z \leftarrow if (Result = 0) then 1, else 0$ 

 $V \leftarrow UNPREDICTABLE$ 

 $C \leftarrow UNPREDICTABLE$ 

#### Description

The logical NOR of the 16-bit words in GPR[Ra] and GPR[Rb] is performed and the result is placed into GPR[Rd].

#### 2.21 LSL

# Logical Shift Left

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1	1	1	1		Rd			Ra		0		im	m4	

## **Syntax**

LSL Rd, Ra, #imm4

eg. LSL R5, R3, #7

## Operation

 $Rd \leftarrow Ra << imm4$ 

 $N \leftarrow if (Result < 0) then 1, else 0$ 

 $Z \leftarrow if (Result = 0) then 1, else 0$ 

 $V \leftarrow UNPREDICTABLE$ 

 $C \leftarrow UNPREDICTABLE$ 

## Description

The 16-bit word in GPR[Ra] is shifted left by the 4-bit amount specified in the instruction, shifting in zeros, and the result is placed into GPR[Rd].

#### LSR 2.22

## Logical Shift Right

#### **Format**

15	1	1	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1		1	0	1		Rd			Ra		0		im	m4	

## **Syntax**

LSR Rd, Ra, #imm4

eg. LSR R5, R3, #7

## Operation

 $Rd \leftarrow Ra >> imm4$ 

 $N \leftarrow if (Result < 0) then 1, else 0$ 

 $Z \leftarrow if (Result = 0) then 1, else 0$ 

 $V \leftarrow UNPREDICTABLE$ 

 $C \leftarrow UNPREDICTABLE$ 

## Description

The 16-bit word in GPR[Ra] is shifted right by the 4-bit amount specified in the instruction, shifting in zeros, and the result is placed into GPR[Rd].

### 2.23 ASR

## Arithmetic Shift Right

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1	1	0	0		Rd			Ra		0		im	m4	

## **Syntax**

ASR Rd, Ra, #imm4

eg. ASR R5, R3, #7

## Operation

 $Rd \leftarrow Ra >>> imm4$ 

 $N \leftarrow \text{if (Result } < 0) \text{ then } 1, \text{ else } 0$ 

 $Z \leftarrow if (Result = 0) then 1, else 0$ 

 $V \leftarrow UNPREDICTABLE$ 

 $C \leftarrow UNPREDICTABLE$ 

## Description

The 16-bit word in GPR[Ra] is shifted right by the 4-bit amount specified in the instruction, shifting in the sign bit of Ra. The result is then placed into GPR[Rd].

2.24 LDW Load Word

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0	0	0	0	0	-	Rd			Ra			i	mm	5	

## **Syntax**

LDW Rd, [Ra, #imm5]

eg. LDW R5, [R3, #7]

## Operation

 $Rd \leftarrow Mem[Ra + imm5]$ 

 $N \leftarrow N$ 

 $\mathbf{Z} \leftarrow \mathbf{Z}$ 

 $V \leftarrow V$ 

 $\mathbf{C} \leftarrow \mathbf{C}$ 

## Description

Data is loaded from memory at the resultant address from addition of GPR[Ra] and the 5-bit immediate value specified in the instruction. The result is then placed into GPR[Rd].

Addressing Mode: Base Plus Offset.

2.25 STW Store Word

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
0	1	0	0	0		Rd			Ra			iı	mm	5	

### **Syntax**

STW Rd, [Ra, #imm5]

eg. STW R5, [R3, #7]

# Operation

 $\text{Mem}[\text{Ra} + \text{imm5}] \leftarrow \text{Rd}$ 

$$N \leftarrow N$$

$$\mathbf{Z} \leftarrow \mathbf{Z}$$

$$V \leftarrow V$$

$$C \leftarrow C$$

# ${\bf Description}$

Data in GPR[Rd] is stored to memory at the resultant address from addition of GPR[Ra] and the 5-bit immediate value specified in the instruction.

Addressing Mode: Base Plus Offset.

#### 2.26 LUI

# Load Upper Immediate

### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	0	1	0	0		Rd					im	m8			

### **Syntax**

LUI Rd #imm8

eg. LUI R5, #93

# Operation

$$Rd \leftarrow \{imm8, 0\}$$

$$N \leftarrow N$$

$$\mathbf{Z} \leftarrow \mathbf{Z}$$

$$V \leftarrow V$$

$$\mathbf{C} \leftarrow \mathbf{C}$$

#### Description

The 8-bit immediate value provided in the instruction is loaded into the top half of GPR[Rd], setting the bottom half to zero. The result is then stored in GPR[Rd].

Addressing Mode: Register-Immediate.

# 2.27 LLI

# Load Lower Immediate

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	0	1	0	1		Rd					im	m8			

### **Syntax**

LLI Rd #imm8

eg. LLI R5, #93

# Operation

$$Rd \leftarrow \{Rd[15:8], imm8\}$$

$$\mathbf{N} \leftarrow \mathbf{N}$$

$$\mathbf{Z} \leftarrow \mathbf{Z}$$

$$\mathbf{V} \leftarrow \mathbf{V}$$

$$\mathbf{C} \leftarrow \mathbf{C}$$

# Description

The 8-bit immediate value provided in the instruction is loaded into the bottom half of GPR[Rd], leaving the top half unchanged. The result is then stored in GPR[Rd].

Addressing Mode: Register-Immediate.

#### 2.28 BR

# **Branch Always**

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1	1	1	0	0	0	0				im	m8			

# Syntax

BR LABEL

eg. BR .loop

# Operation

$$PC \leftarrow PC + imm8$$

$$N \leftarrow N$$

$$Z \leftarrow Z$$

$$V \leftarrow V$$

$$\mathbf{C} \leftarrow \mathbf{C}$$

#### Description

Unconditionally branch to the resultant address from addition of PC and the 8-bit immediate value specified in the instruction. LABEL is a symbolic name for the destination and is capable of jumping forwards or backwards.

#### 2.29 BNE

# Branch If Not Equal

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1	1	1	0	1	1	0				im	m8			

### **Syntax**

BNE LABEL

eg. BNE .loop

## Operation

if (z=0) 
$$PC \leftarrow PC + imm8$$

$$N \leftarrow N$$

$$Z \leftarrow Z$$

$$V \leftarrow V$$

$$C \leftarrow C$$

# Description

Conditionally branch to the resultant address from addition of PC and the 8-bit immediate value specified in the instruction if zero status flag (Z) equals zero. LABEL is a symbolic name for the destination and is capable of jumping forwards or backwards.

#### 2.30 BE

# **Branch If Equal**

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1	1	1	0	1	1	1				im	m8			

### **Syntax**

BE LABEL

eg. BE .loop

## Operation

if 
$$(z=1)$$
 PC  $\leftarrow$  PC + imm8

$$N \leftarrow N$$

$$Z \leftarrow Z$$

$$V \leftarrow V$$

$$\mathbf{C} \leftarrow \mathbf{C}$$

## Description

Conditionally branch to the resultant address from addition of PC and the 8-bit immediate value specified in the instruction if zero status flag (Z) equals one. LABEL is a symbolic name for the destination and is capable of jumping forwards or backwards.

#### 2.31 BLT

# Branch If Less Than

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1	1	1	0	1	0	0				im	m8			

### **Syntax**

BLT LABEL

eg. BLT .loop

# Operation

if  $(n\&!v OR !n\&v) PC \leftarrow PC + imm8$ 

$$N \leftarrow N$$

$$Z \leftarrow Z$$

$$V \leftarrow V$$

$$C \leftarrow C$$

# Description

Conditionally branch to the resultant address from addition of PC and the 8-bit immediate value specified in the instruction if negative status flag and overflow status flag are not equivalent. LABEL is a symbolic name for the destination and is capable of jumping forwards or backwards.

#### **BGE** 2.32

# Branch If Greater Than Or Equal

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1	1	1	0	1	0	1				im	m8			

#### **Syntax**

BGE LABEL

eg. BGE .loop

# Operation

if  $(n\&v OR !n\&!v) PC \leftarrow PC + imm8$ 

$$N \leftarrow N$$

$$Z \leftarrow Z$$

$$V \leftarrow V$$

$$\mathbf{C} \leftarrow \mathbf{C}$$

#### Description

Conditionally branch to the resultant address from addition of PC and the 8-bit immediate value specified in the instruction if negative status flag and overflow status flag are equivalent. LABEL is a symbolic name for the destination and is capable of jumping forwards or backwards.

#### 2.33 BWL

# Branch With Link

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1	1	1	0	0	1	1				im	m8			

#### **Syntax**

BWL LABEL

eg. BWL .loop

#### Operation

$$LR \leftarrow PC + 1$$
;  $PC \leftarrow PC + imm8$ 

$$N \leftarrow N$$

$$Z \leftarrow Z$$

$$V \leftarrow V$$

$$C \leftarrow C$$

# Description

Save the current program counter (PC) value plus one to the link register. Then unconditionally branch to the resultant address from addition of PC and the 8-bit immediate value specified in the instruction.LABEL is a symbolic name for the destination and is capable of jumping forwards or backwards.

2.34 RET Return

# **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1	1	1	0	0	1	0				im	m8			

# Syntax

RET

eg. RET

# Operation

$$\mathrm{PC} \leftarrow \mathrm{LR}$$

$$\mathbf{N} \leftarrow \mathbf{N}$$

$$\mathbf{Z} \leftarrow \mathbf{Z}$$

$$V \leftarrow V$$

$$\mathbf{C} \leftarrow \mathbf{C}$$

# Description

Unconditionally branch to the address stored in the link register (LR).

Addressing Mode: Register-Indirect.

2.35 JMP Jump

# **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1	1	1	0	0	0	1				im	m8			

# Syntax

JMP Ra, #imm5

eg. JMP R3, #7

# Operation

$$PC \leftarrow Ra + imm5$$

$$N \leftarrow N$$

$$\mathbf{Z} \leftarrow \mathbf{Z}$$

$$\mathbf{V} \leftarrow \mathbf{V}$$

$$\mathbf{C} \leftarrow \mathbf{C}$$

# Description

Unconditionally jump to the resultant address from the addition of GPR[Ra] and the 5-bit immediate value specified in the instruction.

Addressing Mode: Base Plus Offset.

#### **PUSH** 2.36

# **Push From Stack**

#### **Format**

	14												
0	1	0	0	1	L	X	X	Ra	0	0	0	0	1

#### **Syntax**

PUSH Ra PUSH LR

eg. PUSH R3 eg. PUSH LR

#### Operation

$$Mem[R7] \leftarrow reg; R7 \leftarrow R7 - 1$$

$$N \leftarrow N$$

$$\mathbf{Z} \leftarrow \mathbf{Z}$$

$$V \leftarrow V$$

$$\mathbf{C} \leftarrow \mathbf{C}$$

# Description

'reg' corresponds to either a GPR or the link register, the contents of which are stored to the stack using the address stored in the stack pointer (R7). This then Decrements the stack pointer by one.

Addressing Modes: Register-Indirect, Postdecrement.

#### 2.37 POP

# Pop From Stack

#### **Format**

		13											
0	0	0	0	1	L	X	X	Ra	0	0	0	0	1

### **Syntax**

POP Ra POP LR eg. POP R3 eg. POP LR

# Operation

$$R7 \leftarrow R7 + 1$$
;  $Mem[R7] \leftarrow reg$ ;  $N \leftarrow N$ 

$$\mathbf{Z} \leftarrow \mathbf{Z}$$

$$V \leftarrow V$$

$$C \leftarrow C$$

# Description

This instruction increments the stack pointer by one. Then 'reg' corresponds to either a GPR or the link register, the contents of which are retrieved from the stack using the address stored in the stack pointer (R7).

Addressing Modes: Register-Indirect, Preincrement.

#### 2.38 **RETI**

# **Return From Interrupt**

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1	0	0	1	0	0	0	1	1	1	X	X	X	X	X

## **Syntax**

RETI

eg. RETI

# Operation

$$PC \leftarrow Mem[R7]$$

$$N \leftarrow N$$

$$\mathbf{Z} \leftarrow \mathbf{Z}$$

$$V \leftarrow V$$

$$\mathbf{C} \leftarrow \mathbf{C}$$

# Description

Restore program counter to its value before interrupt occurred, which is stored on the stack, pointed to by the stack pointer (R7).

Addressing Mode: Register-Indirect.

# 2.39 ENAI

# **Enable Interrupts**

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1	0	0	1	0	0	1	1	1	1	X	X	X	X	X

# Syntax

ENAI

eg. ENAI

# Operation

Set Interrupt Enable Flag

$$\mathbf{N} \leftarrow \mathbf{N}$$

$$Z \leftarrow Z$$

$$\mathbf{V} \leftarrow \mathbf{V}$$

$$\mathbf{C} \leftarrow \mathbf{C}$$

# Description

Turn on interrupts by setting interrupt enable flag to true (1).

#### 2.40 DISI

# Disable Interrupts

# **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1	0	0	1	0	1	0	1	1	1	X	X	X	X	X

# Syntax

DISI

eg. DISI

# Operation

Reset Interrupt Enable Flag

$$N \leftarrow N$$

$$\mathbf{Z} \leftarrow \mathbf{Z}$$

$$V \leftarrow V$$

$$\mathbf{C} \leftarrow \mathbf{C}$$

# Description

Turn off interrupts by setting interrupt enable flag to false (0).

# 2.41 STF

# **Store Status Flags**

#### **Format**

15	14	13	12	11	10	9	8	7	6	5	4	3	2	1	0
1	1	0	0	1	0	1	1	1	1	1	X	X	X	X	X

## Syntax

STF

eg. STF

# Operation

$$Mem[R7] \leftarrow \{12\text{-bit } 0,\,Z,\,C,\,V,\,N\};\,R7 \leftarrow R7 - 1;$$

$$N \leftarrow N$$

$$\mathbf{Z} \leftarrow \mathbf{Z}$$

$$V \leftarrow V$$

$$C \leftarrow C$$

# Description

Store contents of status flags to stack using address held in stack pointer (R7). Then decrement the stack pointer (R7) by one.

 ${\bf Addressing\ Modes:\ Register\text{-}Indirect,\ Post decrement.}$ 

#### 2.42 LDF

# Load Status Flags

#### **Format**

			12												
1	1	0	0	1	1	0	0	1	1	1	X	X	X	X	X

#### **Syntax**

LDF

eg. LDF

# Operation

$$R7 \leftarrow R7 + 1$$

 $N \leftarrow \text{Mem}[R7][0]$ 

 $Z \leftarrow \text{Mem}[R7][3]$ 

 $V \leftarrow \text{Mem}[R7][1]$ 

 $C \leftarrow \text{Mem}[R7][2]$ 

# Description

Increment the stack pointer (R7) by one. Then load content of status flags with lower 4 bits of value retrieved from stack using address held in stack pointer (R7).

Addressing Modes: Register-Indirect, Preincrement.

# 3 Programming Tips

This section gives hints and tips about programming for the Samurai processor.

## 3.1 Branching

The Samurai processor supports four conditional branches. There are **BE**, **BNE**, **BLT** and **BGE**. All conditional branches have an eight bit signed immediate field which is added to the program counter. Labels are supported by the assembler to aid programming (see section 4).

All arithmetic operations update the flags based on the result of the operation. Logic operations update the negative and zero flags. As well as these, the **CMP** and **CMPI** instructions update the flags, but the result is not stored.

Conditional branches should be conducted by first doing a logic or arithmetic operation, followed by the relevant branch instruction. Listing 1 shows assembly for a simple if-then-else clause. First, some definitions are made to make the code more readable. These are discussed further in section 4. A compare is done between the a value and an immediate 1. If these two numbers are not equal, the program flows takes the jump and loads a 0 into b. Else, the program falls through and a 1 is loaded to the b register. The program then takes an unconditional jump to the end of the clause. This is a simple implementation and can be extended to large case statements.

Listing 1: Example code for an *if-then-else* operation.

```
. define a
                 R0
                           ; if 'a' == 1 then b = 1 else b = 0
.define b
                 R1
                            store flags for operation (a - 1)
        CMPI
                 a, #1
        BNE
                  .else
                            branch is a != 1
        LUI
                 b, \#0
                             else fall through
        LLI
                 b, #1
                            load b with 1
        BR
                  . end
.else
        LUI
                 b, #0
                           ; LUI sets lower byte to 0
. end
```

# 3.2 Looping

Listing 2 is an example of how to implement a for loop. Again, definitions are made to give the code more meaning. Then the i variable is initialised to 0 before the loop. Since only greater than or equal, and less than branches are supported, the condition is non-trivial. This is done by using a temporary register which is set to i+1. A compare is done between the temporary register and an immediate 11. This is as  $i \leq 10$  is the same as (i+1) < 11. A **BGE** is done to escape the loop. If this isn't taken, the contents of the loop is executed. i is then incremented and the program jumps to the start of the loop.

Listing 2: Example code for a for loop.

```
define i
                                 ; for (i = 0; i \le 10; i ++)
 . define a
                   R1
                                 ; a = a + i;
 .define temp
                   R2
3
          LUI
                   i,#0
                                   initialise i to 0
 .loop
          ADDI
                   temp, i, #1
                                   need to add one to i
          CMPI
                   temp, #11
                                   check condition
          BGE
                   . end
                                 ; if not(temp >= 11) -> i < 10
          ADD
                   a, a, i
                                 ; do the operation of a += i
          ADDIB
                   i,#1
                                   increment i
          BR
                   .loop
                                   return to the top of the loop
 . end
          . . .
```

Listing 3 describes a *while* loop which doesn't need to be controlled by a loop counter. A variable can completely skip the loop if already larger than the threshold or even remain stuck inside the loop if ill-posed which in this case can be achieved with 0.

Listing 3: Example code for a while loop.

```
; while (i < 15) i = i + i;
. define i
                 R0
.loop
        CMPI
                  i,#15
        BGE
                  . end
                               ; Do op i = i*2
        ADD
                  i,i,i
        BR
                               ; return to the top of the loop
                  .loop
                               ; i = 1 ? 1..2..4..8..16 > out
. end
         . . .
```

## 3.3 Stack Pointer Usage

The Samurai processor uses a full descending stack, example usage is held in listing 4. This means from the initial value the stack pointer is incremented before data is written to memory. It is recommended that the stack pointer is initialised to x where x-1 is the top address in main memory. A **PUSH** increments then writes and a **POP** reads then decrements. Relative loads from the stack pointer are possible and discussed at length in section 3.4.

Listing 4: Example code for **SP** usage.

```
LUI
        SP, #7
LLI
        SP, #208
                        Set SP to 0x07D0
PUSH
        R0
                       R0 written to address 0x07CF
PUSH
         R1
                       R1 written to address 0x07CE
LDW
                       R0 = Mem[0x07CE] (R1 initially)
        R0, [SP, \#0]
LDW
                       R1 = Mem[0x07CF] (R0 initially)
        R1, [SP, #1]
POP
        R1
                       SP moves to 0x07CF
POP
         R0
                      ; SP moves to 0x07D0
```


# 3.4 Sub routine calling convention

Use of a stack frame to pass variables to and from subroutines is recommended when writing assembly for Samurai. This improves code reuse and debugging. Example calls for both linked and unlinked subroutines are contained in listing 5 The stack is explained using figure 2 which shows the state during the main matter of each subroutine.

Subroutine *one* requires two input parameters and returns a single result. The caller pushes two register values to the stack and a third dummy value is created by decrementing the stack pointer. A **BWL** instruction is used to enter the subroutine. The callee now has to save the registers it wishes to work with by storing them on the stack. This subroutine calls another subroutine therefore the first operation should be a **PUSH LR** which places the link register on the stack. Five register values are saved then the two input parameters are loaded relative to the stack pointer. On exit the result is stored at the dummy memory location. Once all register values are restored it is crucial the link register is also restored before the **RET** command is used to return the program to the caller. The caller does one **POP** followed by two dummy pops to return the stack its initial state.

Subroutine two is called from inside subroutine one but because it is a

leaf call then the link register does not have to be copied to the stack. The input parameter is bidirectional and on exit the result is written to the same place from which it was read.



(a) Stack state during (b) Stack state during subsubroutine *one* main routine *two* main matter. matter.

Figure 2: Stack growth.

Listing 5: Example code for calling subroutines.

```
PUSH
                     R0
                                    ; Op1
                                                           STEM CALLER
            PUSH
                                      Op2
                     R1
            SUBIB
                     SP,#1
                                      Dummy push
           BWL
                                      Run Subroutine
                      .one
            POP
                     R0
                                      Result
                     SP, #2
            ADDIB
                                      Duummy pop x 2
            . . .
                                                           CALLEE/CALLER
            PUSH
                     LR
                                    ; Save LR
   .one
            PUSH
                     R0
            PUSH
                     R1
                                    ; Save caller regs
10
            PUSH
                     R2
11
                     R3
           PUSH
12
           PUSH
                     R4
13
                     R3, [SP, #7]
                                    ; R3 - Op2
14
           LDW
           LDW
                     R4, [SP, #8]
                                    ; R4 - Op1
15
            PUSH
                     R3
17
           BWL
                      .two
                                      Nested subroutine
18
           POP
                     R3
                                      Pass and return
19
20
           STW
                     R2, [SP, #6]
                                   ; Output on frame
21
            POP
                     R4
22
                     R3
            POP
23
            POP
                     R2
24
            POP
                     R1
25
            POP
                     R0
26
           POP
                     LR
27
           RET
28
            PUSH
                     R0
                                    ; No LR save
                                                           LEAF CALLEE
   .two
29
           LDW
                     R0, [SP, #1]
30
31
            . . .
           STW
                     R0, [SP, #1]
32
            POP R0
33
           RET
```

# 3.5 Interrupt Service Routines

On reset, interrupts are disabled on the Samurai processor. Two instructions are used to enable and disable interrupts, **ENAI**, **DISI**. These set or clear an internal flag with in the control unit. It is not accessible to the user for reading or branching on it's value. The use of interrupts requires the use of R7 as the stack pointer. The stack pointer should be set up before

interrupts are enabled in the program.

The nIRQ signal to the Samurai processor is an active low, level triggered signal. If the interrupt occurs during an instruction, the instruction is completed before the Interrupt Service Routine (ISR) is entered. The peripheral should hold the nIRQ signal low until it has been cleared by the processor.

Before the ISR is started, the Program Counter value is stored to the stack. Also, interrupts are automatically disabled once an interrupt is triggered to prevent the processor being continually interrupted. Interrupts must be re-enabled before the ISR is completed by the **ENAI** instruction. The first instruction in the ISR **must** be the store flags instruction, **STF**. The final two instructions in the ISR **must** be load flags **LDF** and return from interrupt **RETI**. The user is responsible for saving all the registers and restoring them before returning.

Nested interrupts are supported on the Samurai if required. The initial interrupt must first be cleared. Interrupts can then be re-enabled. If a new interrupt occurs, the ISR is run. Once the second ISR is completed, the program flow is returned to where it was before hand and the first ISR run is then complete.

The ISR can also conduct a function call. However, this is not recommended as the ISR should be short in length.

The general outline for the ISR is:

- 1. Store Flags
- 2. Push registers to stack
- 3. Clear interrupt source
- 4. Enable interrupts
- 5. Process data
- 6. Restore registers
- 7. Load Flags
- 8. Return from Interrupt

The ISR is implemented by using the ".isr" or ".ISR" label. It can be placed anywhere in the code and be any length. A general outline in assembly language is shown in listing 6. This structure should be followed for the ISR.

Listing 6: Example outline for the Interrupt Service Routine

1		LUI	R7, #7	;	set up stack pointer
2		$_{ m LLI}$	R7, #208		
3		ENAI		;	enable the interrupts
4		BR	$.  \mathrm{main}$	;	go to main
5	.isr	STF		;	store flags
6		PUSH	R0	;	free some registers to use
7		PUSH	R1		
8		PUSH	R2		
9				;	clear interrupt source
10		ENAI		;	enable interrupts
11				;	process data if necessary
12		POP	R2	;	restore the regs in reverse order
13		POP	R1		
14		POP	R0		
15		LDF		;	load the flags
16		RETI		;	end of the ISR

# 4 Assembler

The current instruction set architecture includes an assembler for converting assembly language into hexadecimal. This chapter outlines the required formatting and available features of this assembler.

# 4.1 Instruction Formatting

Each instruction must be formatted using the following syntax. Here " $[\dots]$ " indicates an optional field:

```
[.LABELNAME] MNEMONIC, OPERANDS, ..., : [COMMENTS]
```

For example:

```
.loop ADDI, R5, R3, #5 : Add 5 to R3
```

Comments may be added by preceding them with either: or;

Accepted general purpose register values are: R0, R1, R2, R3, R4, R5, R6, R7, SP. These can be upper or lower case and SP is equivalently evaluated to R7.

Branch instructions take a symbolic reference to the destination. Each type of branch supports moving up to 127 lines forward, or 128 lines backwards. But if a branch is over this limitation, the assembler will automatically create additional instructions to enable greater distances. Each additional branch added will cause two more lines of code to be added to the outputted file.

All label names must begin with a '.' while .ISR/.isr and .define are special cases used for the ISR and variable definitions respectively.

Instruction-less or comments only lines are allowed within the assembly file.

#### Special Case Label

The .ISR/.isr label is reserved for the Interrupt Service Routine and may be located anywhere within the file but must finish with a **RETI** instruction. There is no restriction on size of the service routine. Branches may occur within the ISR, but are not allowed into this service routine with the exception of a return from a separate subroutine. For ISR set-up code to be added successfully, there must be at least one unused register within the first 11 lines of the program file, excluding .define statements and the ISR.

#### 4.2 Assembler Directives

Symbolic label names are supported for branch-type instructions. Following the previous syntax definition for '.LABELNAME', they can be used instead of numeric branching provided they branch no further than the maximum distance allowed for the instruction used. Definitions are supported by the assembler. They are used to assign meaningful names to the GPRs to aid with programming. Definitions can occur at any point within the file and create a mapping from that point onwards. Different names can be assigned to the same register, but only one is valid at a time.

The accepted syntax for definitions is:

.define NAME REGISTER

# 4.3 Running The Assembler

The assembler is a python executable and is run by typing "./assemble.py". Alternatively, the assembler can be placed in a folder on the users path and executed by running "assemble.py". It supports Python versions 2.4.3 to 2.7.3. A help prompt is given by the script if the usage is not correct, or given a -h or --help argument.

By default, the script will output the assembled hex to a file with the same name, but with a '.hex' extension in the same directory. The user can specify a different file to use by using a -o filename.hex or --output=filename.hex argument to the script. The output file can also be a relative or absolute path to a different directory.

The full usage for the script is seen in listing 7. This includes the basic rules for writing the assembly language and a version log.

Listing 7: Assembler help prompt

```
Usage: assemble.py [-o outfile] input
    -Team R4 Assembler Help---
       -Version: 1 (CMPI addition onwards)
                 2 (Changed to final ISA, added special case I's
     and error checking
                 3 (Ajr changes - Hex output added, bug fix)
                 4 (Added SP symbol)
                 5 (NOP support added, help added)
                 6 (Interrupt support added [ENAI, DISI, RETI])
                 7 (Checks for duplicate Labels)
11
                 8 (Support for any ISR location & automated
     startup code entry)
                 9 (Support for .define)
13
                 10 (Changed usage)
      11 (ISR setup shortened, Numeric branching support removed)
      12 (Branches automatically extended if out of 8-bit range)
      13 (Comments in hexfile)
17
        Current is most recent iteration
  Input Syntax: ./assemble filename
  Commenting uses : or ;
20
  Labels start with '.': SPECIAL .ISR/.isr-> Interrupt Service
21
     Routine)
                          SPECIAL .define -> define new name for
     General Purpose Register, .define NAME R0-R7/SP
  Instruction Syntax: .[LABELNAME] MNEUMONIC, OPERANDS, ..., :[
     COMMENTS]
  Registers: R0, R1, R2, R3, R4, R5, R6, R7=SP
  Branching: Only Symbolic Supported
25
26
27
  Notes:
         Input files are assumed to end with a .asm extension
         Immediate value sizes are checked
         Instruction-less lines allowed
         .ISR may be located anywhere in file
31
         . define may be located anywhere, definition valid from
32
     location in file onwards, may replace existing definitions
     Error message line numbers are prefixed with f for assembly
33
      file and p for preprocessed code
34
36 Options:
   --version
                           show program's version number and exit
```

```
-h, --help show this help message and exit
-o FILE, --output=FILE
output file for the assembled output
```

# 4.4 Error Messages

This is a list of all the error messages produced by the assembler. Each time an error is thrown, the error number, a brief description and the line it occurred on is displayed before exiting. A 'f' corresponds to a line number in the assembly file, and a 'p' corresponds to the pre-processed code list displayed on screen. Figure 3 shows an example screen shot of an error message.

```
hind<~/VLSI/GIT/Design/InstructionSet/Assembler>$ ./assemble.py testprog
------Converting File testprog.asm-----
----Interpreting Syntax------
ERROR1: Unrecognised Mneumonic lineNo: f37 -> ADDX_R6 R6 5
```

Figure 3: Error1 Example Screen Shot

Code	Description
ERROR1	Instruction mnemonic is not recognised
ERROR2	Register code within instruction is not recognised
ERROR3	Branch condition code is not recognised
ERROR4	Attempting to branch to undefined location
ERROR5	Instruction mnemonic is not recognised
ERROR6	Attempting to shift by more than 16 or perform a negative shift
ERROR7	Magnitude of immediate value for ADDI, ADCI, SUBI, SUCI, LDW, STW, CMPI or JMP is too large
ERROR8	Magnitude of immediate value for ADDIB, SUBIB, LUI or LLI is too large
ERROR9	Attempting to jump more than 127 forward or 128 backwards
ERROR10	Duplicate symbolic link names
ERROR11	Illegal branch to ISR
ERROR12	Multiple ISRs in file
ERROR13	Invalid formatting for .define directive
ERROR14	Could not find empty register in first 10 lines for automated ISR setup
ERROR15	Instruction does not have enough operands

# 5 Programs

Every example program in this section uses R7 as a stack pointer which is initialised to the by the program to 0x07D0. The simulation environment contains an area of an area of memory with 2048 locations and memory mapped deices. There are 16 switches at location 0x0800, 16 LEDs at location 0x0801 and a serial I/O device which can be read from location 0xA000 and has a control register at location 0xA001.

# 5.1 Multiply

The code for the multiply program is held in Appendix A.1 listing 14. A sixteen bit number is read from input switches, split in to lower and upper bytes which are then multiplied. The resulting sixteen bit word is written to the LEDs before reaching a terminating loop. Equation (1) formally describes the algorithm disregarding limitations.

$$A = M \times Q = \sum_{i=0}^{\infty} 2^{i} M_{i} Q \text{ where } M_{i} \in \{0, 1\}$$
 (1)

The subroutine operation is described in listing 8, using C. If the result is greater than or equal to 2<sup>16</sup> the subroutine will fail and return zero. The lowest bit of the multiplier controls the accumulator and the overflow check. The multiplier is shifted right and the quotient is shifted left at every iteration. An unconditional branch is used to keep the algorithm in a while loop. The state of the multiplier is compared at every iteration against zero when the algorithm is finished. As size of the multiplier controls the number of iterations a comparison is made on entry to use the smallest operand.

Listing 8: Multiply Subroutine

```
uint16_t multi(uint16_t op1, op2)
      uint16_t A,M,Q;
      A = 0;
                                      // Make M small, less loops
      if(op1 < op2)
           M = op1; Q = op2;
      } else {
          M = op2; Q = op1;
                                      // No loop counter
      while (1) {
           if (M & 0x0001) {
                                         LSb
               A = A + Q;
11
               if(A > 0xFFFF)
                                      // Using carry flag
12
                    return 0;
                                      // Overflow - fail
14
15
           M = M >> 1;
           if(0 == M)
                                      // Finished - pass
               return A;
18
19
           if (Q & 0x8000) {
20
                                      // Q >= 2^16 - fail
               return 0;
21
           Q = Q << 1;
23
      }
24
25
```

#### 5.2 Factorial

The code for the factorial program is held in Appendix A.2 listing 15. It is possible to calculate the factorial of any integer value between 0 and 8 inclusive. The subroutine is called which in turn calls the multiply subroutine discussed in section 5.1. The factorial subroutine does no parameter checking but the multiply code does so if overflow does occur zero is propagated and returned; zero is not a possible factorial. The result is calculated recursively as described using C in listing 9. Large values can cause stack overflow the main body of code makes sure inputs, read from the switches, are sufficiently small.

Listing 9: Recursive Factorial Subroutine

### 5.3 Random

The code for the random program is held in Appendix A.3 listing 16. A random series of numbers is achieved by simulating the 16 bit linear feedback shift register in Figure 4. This produces a new number every 16 sixteen clock cycles so in this case a simulation subroutine is called 16 times. A seed taken from switches and passed to the first subroutine call via the stack is altered and passed to the next subroutine call. No more stack operations are performed. A load from the stack pointer is used write a new random number to LEDs. All contained within an unconditional branch but a loop counter is used control write and reset.




Figure 4: 16 Bit Linear Feedback Shift Register.

A two input XOR gate is simulated using the XOR operation along with shifting to compare bits in different locations. Bits 2 and 4 are used as inputs so a logical shift left by two is used to align them at the bit 4 position. Masking the output value is used feedback to the top bit. A seed can be any value expect 0x0000 as this will not produce a logical 1 for feedback therefore remaining stuck in the same state. This is described using C in listing 10.

Listing 10: Linear Feedback Shift Register Subroutine

# 5.4 Interrupt

The code for the interrupt program is held in Appendix A.4 listing 17. This is the most complex example and makes use of both the multiply and factorial subroutines in sections 5.1 and 5.2 respectively. The interrupt services a serial device writing input data to a 4 byte circular buffer. A main program checks to see if data is in the buffer then and if so calculates the factorial writing the result to the LEDs. The buffer is purposefully small to test overflow.

The functionality is explained using in listing 11. Main compares the read and write pointers in loop because only when the two are different does the buffer contain data. If the two are different then the read pointer is incremented at which point it will wrap round to the beginning of the buffer. The new value for the read pointer is written back to memory and the factorial of the data is calculated.

Listing 11: Serial Device Interrupt Service Request

```
1 #define TOP
                    0x0206
2 #define BOTTOM
                    0x0202
3 #define WRITE
                    0x0201
4 #define READ
                    0x0200
5 #define SERIAL
                    0xA000
  #define LEDS
                    0x0801
  isr(){
       uint16_t data, readPtri, writePtr;
       data = read(SERIAL);
                                      // Don't lose event
      asm("ENAI");
                                      // nested ints
11
       readPtr = read(READ);
12
       writePtr = read(WRITE);
13
       if(((readPtr-1) = writePtr)
14
           (readPtr
                        == BOTTOM)
                                           (writePtr
                        == (TOP-1)
                                          ) {
16
                                         full, don't write
           return
17
18
19
       if (readPtr == BOTTOM)
       write(readPtr, data);
                                      // write to buffer
20
       writePtr++;
21
       if(writePtr == TOP){
22
           writePtr = BOTTOM;
23
       }else{
24
           writePtr++;
25
26
       write(WRITE, writePtr);
27
       return
28
  }
29
30
  void main(){
31
       uint16_t readPtr, writePtr, data;
32
      do{
33
           readPtr = read(READ);
                                     // keep checking buffer
34
           writePtr = read (WRITE);
35
       } while (readPtr == writePtr)
36
       data = read(readPtr);
37
       readPtr++;
38
       if(readPtr == TOP)
           readPtr = BOTTOM;
40
41
       write (READ, readPtr);
                                     // Write new read ptr
42
       write (fact (data));
43
44
```

## 6 Simulation

#### 6.1 Running the simulations

A python script, sim.py, was written to automatically invoke the assembler and simulator. The passed program is only assembled if the file exists with an extension of .asm. This allows for raw hex to be passed to the simulator where necessary. If a .hex file is passed, and a .asm file exists of the same name, the assembler will be invoked. The sim.py script is configured to be run from within the assembler directory.

The usage for the script is:

```
sim.py [-t type] [-m module.sv / -p program.asm ] [ -s
switchvalue ] [ -gdS ] [+define+extra_definitions]
```

All simulation types are supported. As well as full system simulations, the sim.py script also allows for other testbenches to be run. All stimulus files are maintained in a directory and the testbenches can be run on verilog or magic modules. Where a Magic design is to be simulated, the script automatically extracts the netlist. This is done to prevent the Magic design and netlist being inconsistent.

The sim.py script provides a help prompt when run with -h or --help arguments. The help prompt is also displayed when incorrect arguments are supplied. The full help prompt is show in listing 12.

By default, the graphical user interface is not invoked. This can be done with the -g or --gui tags. A debug option, -d, exists when the user wants to get the majority of the simulation command, but modify it slightly.

The program and module options should never be defined at the same time. One of them, however, should be. The program option is assembled, if necessary, and defined in the simulation command. The module option checks for the testbench file (identified by *module\_stim.sv*) within the verification folder. The testbench is then used as the top level module.

The type of simulation can be any of the folders in the verilog directory, for example behavioural, mixed or extracted. A special type, magic can be used. When this is done so, the magic folder, /design/fcde/magic/design, is checked for the module given. Type magic and a program is equivalent to an extracted type simulation and is treated as such. The type is also given as a definition to the simulator, allowing reuse of test benches.

The value of the switches can be easily defined by using the -s tag. The value given after this option is then passed to the simulator as a definition. If other definitions are required (for example, the serial data file), they can be defined, in full, in the trailing arguments. All trailing arguments are appended to the simulation command, allowing for the user to customise the invocation beyond the scope of the script.

A scan path simulation can also be run. This is done by running ./sim.py -S and allows the same use described above for invoking the GUI. If the -S option is defined, any program or module also given is ignored. The scan path test pulses a signal on the SDI line, and verifies a pulse is seen on the output. The clock cycles, and therefore the number of registers, are counted and reported upon success of the simulation.

Listing 12: Help prompt for the sim.py script.

```
Usage: sim.py [-t type] [-m module.sv / -p program.asm ]
     switchvalue | [-gdS] [+define+extra_definitions]
  trailing arguments are given to the simulator directly
  Options:
                           show program's version number and exit
    --version
    -h, --help
                           show this help message and exit
    -m MODULE, ---module=MODULE
                           module to simulate - should not be
      defined if program
                           is
    -t TYPE, --type=TYPE
                           Type of simulation to run, e.g.
     behavioural (default),
                           mixed, extracted, magic
12
    -p PROGRAM, --prog=PROGRAM
13
                           program to run should not be defined if
14
     module is. Hex
                           or ASM can be passed. ASM files will be
     assembled
                           before running the simulator.
    -H HOME, --home=HOME
                           Use a custom home directory
17
    -g, --gui
                           Run the simulation with a GUI
18
    -s SWITCHES, --switches=SWITCHES
19
                           Value of switches to pass to the
20
      simulation
                           Make, but don't execute, the command
21
    -S, --scanpath
                           Run the scan path simulation
```

Finally, a -H or --home tag exists to override the default expected location. The script expects to be in the assembler directory within the verilog folder. If an absolute path is passed, the script will use this as the base directory with the *behavioural*, *mixed* and *extracted* folders in. A relative path can be passed. This should be the path from the folder the script is run from to the verilog directory.

#### 6.2 Serial Data

The serial data file used is located in the programs directory. This is a hex file with white space separated values of the form "time data". Both values are given as hexadecimal numbers. The data is then sent at the time to the processor by the serial module. An example serial data hex file is shown in listing 13.

Listing 13: Example serial data file

```
// Hex file to specify serial data input
  248
         7
  48F
         6
  6D6
         5
         4
  91D
         7
 B64
 DAB
         5
  36B1
           3
           2
  6D61
```

#### 6.3 Run Time

The number of clock cycles for each program to fully run is shown in table 1. Factorial run time is given for an input of 8 and is the worst case. Random is the time taken to compute a new value of the pseudo-random sequence. Interrupt is dependant on the serial data input and the time is given for the serial data file mentioned above.

#### 6.4 Simulation

A dissembler is also implemented in System Verilog to aid debugging. It is an ASCII formatted array implemented at the top level of the simula-

Table 1: Clock cycles required for each program to run

Program	Clock Cycles
Multiply	1,100
Factorial	5,700
Random	2,500
Interrupt	30,000

tion. It is capable of reading the instruction register with in the design, and reconstructing the assembly language of the instruction and is supported in behavioural, mixed and extracted simulations. It will show the opcode, register addresses and immediate values. It is automatically included by the TCL script. The TCL script also opens a waveform window and adds important signals.

## A Code Listings

All code listed in this section is passed to the assembler as is and has been verified using the final design of the processor.

## A.1 Multiply

Listing 14: multiply.asm

```
LUI
                     SP, #7
                                         ; Init SP
                     SP, #208
            LLI
                     R3, #8
                                         ; SWs addr
            LUI
                     R0, [R3, #0]
           LDW
                                         ; READ SWs
            LUI
                     R1, #0
            LLI
                     R1, #255
                                        ; 0x00FF in R1
           AND
                     R1, R0, R1
                                          Lower byte SWs in R1
           LSR
                     R0, R0, #8
                                          Upper byte SWs in R0
           PUSH
                     R0
                                          Op1
           PUSH
                     R1
                                          Op2
           PUSH
                     R2
                                          Place holder is zero reset
           BWL
                     . multi
                                           Run Subroutine
12
           POP
                     R1
                                           Result
13
                     SP, #2
            ADDIB
                                          Duummy pop
14
                                           Address of LEDS
15
            ADDIB
                     R3, #1
           STW
                     R1, [R3, \#0]
                                           Result on LEDS
  . end
           BR
                     . end
                                           Finish loop
17
                     R0
  . define M
  .define Q
                     R1
  .define A
                     R2
21
  .define i
                     R3
  . multi
           PUSH
                     Μ
           PUSH
                     Q
23
           PUSH
                     Α
24
           PUSH
                     i
25
           LDW
                     M, [SP, #5]
                                          Off stack frame
26
           LDW
                     Q, [SP, #6]
                     i ,M,#15
           LSL
  . mloop
                                          Bit one only
28
                     i,#0
           CMPI
29
           BE
                                        M[1] != 1
                     . nAcc
30
                                        ; A = A + Q
           ADD
                     A, A, Q
  . nAcc
            LSR
                     M, M, #1
                                        M = M >> 1
32
           CMPI
                     M, \#0
33
           BE
                     . done
34
            \operatorname{LSL}
                                         ; Q = Q << 1
                     Q, Q, \#1
```

```
BR
                       . mloop
36
                       A, [SP, #4]
   . done
            STW
                                           ; Res on stack frame
37
            POP
                       i
38
            POP
                       Α
39
            POP
                       Q
40
            POP
                      Μ
41
            RET
42
```

#### **Factorial A.2**

Listing 15: factorial.asm

```
. define SW
                     R0
  . define LEDS
                     R1
  . define data
                     R3
           LUI
                     SP, #7
                     SP, #208
           LLI
                     SW, #8
           LUI
                                         Address of switches
                     LEDS,SW,#1
           ADDI
                                         Address of LEDS
  .start
           LDW
                     data, [SW, \#0]
                                        ; Read switches into R1
                                          Pass para
           PUSH
                     data
                     . fact
           BWL
                                          Run Subroutine
10
           POP
                                         Para overwritten with result
                     data
11
12
           STW
                     data, [LEDS, #0]
                                        ; Result on LEDS
           BR
                                        ; Do it again
                     .start
  . define Op
                     R0
14
   define min
                     R1
15
  . fact
                     LR
           PUSH
16
           PUSH
                     Op
17
           PUSH
                     \min
18
           LDW
                     Op, [SP, #3]
                                   ; Get para
19
           CMPI
                     Op, \#0
20
           BE
                     .retOne
                                   ; 0! = 1
21
           SUBI
                     \min, Op, #1
22
           PUSH
                     \min
                                   ; Pass para
23
           BWL
                     . fact
                                     The output remains on the stack
24
           PUSH
                     Op
                                     Pass para
25
           SUBIB
                     SP,#1
                                     Placeholder
26
           BWL
                     . multi
27
           POP
                     Op
                                     Get res
28
           ADDIB
                     SP,#2
                                   ; pop x 2
29
                     Op, [SP, #3]
           STW
  .out
30
           POP
                     min
31
           POP
                     Op
32
```

```
POP
                      LR
33
            RET
   .retOne ADDIB
                      {\rm Op},\#1
                                   ; Make it 1
35
            BR .out
  .define M
                      R0
37
  .define Q
                      R1
38
  . define A
                      R2
39
  .define i
                      R3
   . multi
            PUSH
                     Μ
41
            PUSH
                      Q
42
            PUSH
                      Α
43
            PUSH
                      i
44
            LDW
                      M, [SP, #5]
                                          ; Off stack frame
                      Q, [\,SP, \#6]
            LDW
46
            SUB
                      A, A, A
47
  . mloop
            LSL
                      i ,M,#15
                                          ; Bit one only
48
            CMPI
                      i ,#0
                                          M[1] != 1
            BE
                      . nAcc
50
            ADD
                      A, A, Q
                                          ; A = A + Q
  . nAcc
            LSR
                      M, M, #1
                                          M = M >> 1
52
            CMPI
                      M, \#0
53
            BE
                      . done
54
            LSL
                                          ; Q = Q << 1
                      Q, Q, \#1
55
            BR
                      . mloop
56
            STW
                      A, [SP, #4]
                                         ; Res on stack frame
   . done
57
            POP
                      i
58
            POP
                      A
59
            POP
                      Q
60
                     Μ
            POP
61
            RET
```

## A.3 Random

#### Listing 16: random.asm

```
LUI
                   SP, #7
                                 ; Init SP
                   SP, #208
          LLI
                   R0,#8
                                 ; SW Address in R0
          LUI
                   R0,#0
          LLI
          LDW
                   R1, [R0, \#0]
                                 ; Read switches into R1
          ADDIB
                   R0, #1
                                 ; Address of LEDS in R0
          PUSH
                   R1
          SUB
                   R4, R4, R4
                                 ; Reset Loop counter
 .reset
          BWL
9 . loop
                    . rand
```

```
CMPI
                      R4, #15
10
            BE
                       .write
11
            ADDIB
                                      ; INC loop counter
                      R4, #1
12
            BR
                       .loop
13
                      R1, [SP, \#0]
   .write
            LDW
                                      ; No pop as re-run
14
                      R1, [R0, #0]
            STW
                                      ; Result on LEDS
15
            BR
                       .\,\mathrm{reset}
16
   . rand
            PUSH
                      R0
                                      ; LFSR Sim
17
            PUSH
                      R1
                                        Protect regs
18
            PUSH
                      R2
19
            LDW
                      R0, [SP, #3]
                                      ; Last reg value
20
            LSR
                      R0, R0, #1
                                      ; Shifted reg
21
                      R1, R0, R1
            XOR
                                      ; xor 0 and 1
22
            LUI
                      R2, \#0
23
            LLI
                      R2, #1
24
            AND
                      R1, R2, R1
                                      ; Mask off Bit 0
25
            CMPI
                      R1, \#0
26
            BE
                       . done
27
            LSL
28
                      R1, R2, \#15
                                      ; or with 0x8000
                      R0, R0, R1
            OR
29
   . done
            STW
                      R0, [SP, #3]
30
            POP
                      R2
31
            POP
                      R1
32
            POP
                      R0
33
            RET
34
```

## A.4 Interrupt

Listing 17: interrupt.asm

```
. define addr
                     R0
  . define data
                     R1
           LUI
                     SP,#7
                     SP, #208
           LLI
           LUI
                     addr, #2
                                         ; read ptr
                                                       0x0200
           ADDI
                     data, addr, #2
                                        0x0202
           STW
                     data, [addr,#0]
                                        ; Read ptr set to
                                                                0x0202
                     \mathrm{data}\;,[\;\mathrm{addr}\;,\#1]
           STW
                                        ; Write ptr set to
                                                                0x0202
           LUI
                     addr, #160
                                         ; Address of Serial control reg
           LLI
                     addr, #1
10
           LUI
                     data,#0
           LLI
                     data,#1
                                         ; Data to enable ints
12
           STW
                     data, [addr,#0]
                                        ; Store 0x001 @ 0xA001
13
           ENAI
                                         ; Interrupts on
14
```

```
. main
            LUI
                     R0, #2
                                    ; Read ptr address in R0
15
                     R0, #0
            LLI
           LDW
                     R2, [R0, #0]
                                      Read ptr in R2
17
           LDW
                     R3, [R0, #1]
                                    ; Write ptr in R3
18
           CMP
                     R2,R3
19
            BE
                                    ; Jump back if the same
                      . main
20
           LDW
                     R3, [R2, \#0]
                                    ; Load data out of buffer
21
            ADDIB
                     R2, #1
                                    ; Inc read ptr
22
                     R0, R0, R0
            SUB
23
                     R0,#2
24
            LUI
            LLI
                     R0,#6
25
            SUB
                     R0, R0, R2
26
            BNE
                      . wrapR
27
            SUBIB
                     R2, #4
28
                                    ; Read ptr address in R0
   . wrapR
           LUI
                     R0, #2
29
            LLI
                     R0, #0
30
           STW
                     R2, [R0, \#0]
                                   ; Store new read pointer
            SUB
                     R4, R4, R4
32
                     R4, #15
            LLI
            AND
                     R3, R4, R3
34
            CMPI
                     R3, #8
35
            BE
                      . do
36
            LLI
                     R4, #7
37
            AND
                     R3, R3, R4
38
            PUSH
   . do
                     R3
39
           BWL
                      . fact
40
            POP
                     R3
41
            LUI
                     R4, #8
42
                                    ; Address of LEDs
            LLI
                     R4, #1
43
           STW
                                    ; Put factorial on LEDs
                     R3, [R4, \#0]
44
            BR
                      . main
                                         ; look again
45
  . define Op
                     R0
  .define min
                     R1
47
   . fact
            PUSH
                     LR
48
            PUSH
                     Op
49
            PUSH
                     \min
           LDW
                     Op, [SP, #3]
                                    ; Get para
51
            CMPI
                     Op, \#0
52
            BE
                                    ; 0! = 1
                      .retOne
53
            SUBI
                     \min, Op, \#1
54
            PUSH
                                      Pass para
                     \min
56
           BWL
                      . fact
                                      The output remains on the stack
            PUSH
                     Op
                                      Pass para
57
            SUBIB
                     SP,#1
                                      Placeholder
58
           BWL
                      . multi
59
```

```
POP
                      Op
                                     ; Get res
60
                      SP,#2
            ADDIB
                                     ; pop x 2
61
            STW
                      Op, [SP, #3]
   out.
62
            POP
                      \min
63
            POP
                      Op
64
            POP
                      LR
65
            RET
66
   .retOne ADDIB
                      Op, #1
                                     ; Make it 1
67
            BR .out
68
   .define M
                      R0
69
   . define Q
                      R1
70
   . define A
                      R2
71
   . define i
                      R3
72
   . multi
            PUSH
                      Μ
73
                      Q
            PUSH
74
            PUSH
                      Α
75
            PUSH
76
            LDW
                      M, [SP, #5]
                                          ; Off stack frame
77
                      Q, [SP, #6]
78
            LDW
            SUB
                      A, A, A
79
   . mloop
            LSL
                      i ,M,#15
                                          ; Bit one only
80
            CMPI
                      i,#0
81
                                          M[1] != 1
            BE
                       . nAcc
82
                                          ; A = A + Q
            ADD
                      A, A, Q
83
                                          M = M >> 1
   . nAcc
            LSR
                      M, M, \# 1
84
            CMPI
                      M, \#0
85
            BE
                       . done
86
            LSL
                      Q, Q, #1
                                          Q = Q << 1
87
            BR
                       . mloop
88
                      A, [SP, #4]
                                          ; Res on stack frame
   . done
            STW
89
            POP
90
            POP
                      Α
91
            POP
                      Q
92
                      Μ
            POP
93
            RET
94
                                     ; Keep flags, disable auto
            STF
95
   . isr
            PUSH
                      R0
                                     ; Save only this for now
96
                      R0, #160
            LUI
97
            LLI
                      R0,#0
98
            LDW
                      R0, [R0, \#0]
                                     ; R1 contains read serial data
99
            ENAI
                                     ; Don't miss event
100
101
            PUSH
                      R1
            PUSH
                      R2
102
            PUSH
                      R3
103
            PUSH
                      R4
104
```

```
LUI
                      R1,#2
             LLI
                      R1,#0
106
            LDW
                      R2, [R1, \#0]
                                    ; R2 contains read ptr
            ADDI
                      R3, R1, #1
108
            LDW
                      R4, [R3, \#0]
                                     ; R4 contain the write ptr
109
                                     ; Get out if W == R - 1
            SUBIB
                      R2, #1
110
            CMP
                      R4, R2
111
            BE
                      . isrOut
112
            ADDIB
                      R2,\#1
113
             LUI
                      R1,#2
114
             LLI
                      R1,#2
115
            CMP
                      R2,R1
116
            BNE
                       .write
117
            ADDIB
                      R1, #3
118
            CMP
                      R4,R1
119
            BE
                       .isrOut
120
            STW
                      R0, [R4, \#0]
                                    ; Write to buffer
   . write
121
             ADDIB
                      R4, #1
122
             LUI
                      R1,#2
123
             LLI
                      R1,#6
124
            CMP
                      R1, R4
125
            BNE
                      . wrapW
126
                      R4, #4
            SUBIB
127
            STW
   . wrapW
                      R4, [R3, \#0]; Inc write ptr
128
   .isrOut POP
129
                      R4
            POP
                      R3
130
                      R2
            POP
131
            POP
                      R1
132
            POP
                      R0
133
            LDF
134
            RETI
135
```

# B Declaration of Work

Table 2 shows the break down of the work for this programming guide. Where multiple team members contributed to a section, approximate work percentages are given.

Table 2: Work Breakdown for the Programmers Guide

Chapter	User	
Introduction	hl13g10	
Register Description	arr1g13 (50%), hl13g10 (50%)	
Instruction Set	mw20g10	
Programming Tips	ajr2g10 (50%), hl13g10 (50%)	
Assembler	mw20g10 (75%), hl13g10 (50%)	
Programs	ajr2g10	
Simulation	hl13g10	
Code Listings	ajr2g10	